Уравнения Пенлеве

К лекции 12 (2024)

---

1 Галилеевская инвариантность и P_1

(1) $u_t = u_{xxx} - 6\,u\,u_x$

КдФ допускает решения вида $u(x,\,t) = 2\,w(z) + 2\,t,\;\; z = x - 6\,t^2,\;\; w'' = 6\,w^2 + z$

In[106]:=

```
ndsol[w0_, w1_] := NDSolve[{w''[z] == 6 w[z]^2 + z, w[0] == w0, w'[0] == w1,
    WhenEvent[Abs[w[z]] > 10, "StopIntegration"]}, w, {z, -35, 5}][[1]]

pl[s_] := Plot[w[z] /. s, {z, -25, s[[1, 2, 1, 1, 2]]},
  PlotRange → {{-23.5, 3.25}, {-3.35, 4.75}},
  PlotStyle → Blue,
  AxesLabel →
   (Style[#, {FontSize → 20, FontFamily → "Times New Roman", FontSlant → Italic}] & /@
     {"z", "w"}),
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 400
 ]

Manipulate[s = ndsol[w0, w1];
 pl[s],
 {{w0, 0.9}, -3, 3, Appearance → "Labeled"},
 {{w1, 1.7}, -3, 3, Appearance → "Labeled"}
]
```

Out[108]=



График строится медленно (можно уменьшить PlotPoints)

In[109]:=

```
g = Plot3D[2 w[x - 6 t^2] + 2 t /. s, {x, -15, 5}, {t, -2, 2},
  PlotRange → {{-15, 5}, {-1.7, 1.7}, {-15, 6}},
  PlotPoints → 250,
  Mesh → None,
  ClippingStyle → None,
  AxesLabel → Evaluate[Style[#, Italic, 24] & /@ {"x", "t", "u"}],
  ImageSize → 600,
  BaseStyle → {FontFamily → "Times", FontSize → 16}
 ]
```

InterpolatingFunction: Input value {-38.9995} lies outside the range of data in the interpolating function. Extrapolation will be used.

Out[109]=



2 Группа, отвечающая симметри Галилея+сдвиг *t*

Пояснение, откуда взялась эта редукция. Стационарное уравнение для симметрии: $c\,(6\,t\,u_x - 1) + u_t = 0$. Решаем его методом характеристик, находим первые интегралы.

In[110]:=

```
Clear[u]

dif[f_] := 6 c t D[f, x] + D[f, t] + c D[f, u]
C1 = x - 3 c t^2;
C2 = u - c t;

dif[C1]
dif[C2]
U = c t + F[C1];
Expand[c (6 t D[U, x] - 1) + D[U, t]]
```

Out[114]=

```
0
```

Out[115]=

```
0
```

Out[117]=

```
0
```

Уравнения для характеристик фактически совпадают с системой Ли, ее решение с тождественными начальными условиями дает групповое преобразование.

In[118]:=

```
DSolve[{x'[a] == 6 c t[a], t'[a] == 1, u'[a] == c,
   x[0] == x0, t[0] == t0, u[0] == u0},
  {x[a], t[a], u[a]}, a]
```

Out[118]=

```
{{t[a] → a + t0, x[a] → 3 a^2 c + 6 a c t0 + x0, u[a] → a c + u0}}
```

По этому решению определяем преобразование и проверяем, что это действительно однопараметрическая группа.

In[119]:=

```
G[a_, {x_, t_, u_}] := {3 a^2 c + 6 a c t + x, t + a, u + a c}
G[0, {x, t, u}]
G[a, G[b, {x, t, u}]]
G[a + b, {x, t, u}]
Expand[%% - %]
```

Out[120]=

```
{x, t, u}
```

Out[121]=

```
{3 a^2 c + 3 b^2 c + 6 b c t + 6 a c (b + t) + x, a + b + t, a c + b c + u}
```

Out[122]=

```
{3 (a + b)^2 c + 6 (a + b) c t + x, a + b + t, (a + b) c + u}
```

Out[123]=

```
{0, 0, 0}
```

Первые интегралы – инварианты группы.

In[124]:=

```
r = {x → 3 a^2 c + 6 a c t + x, t → t + a, u → u + a c};
C1 /. r
Expand[% - C1]
C2 /. r
Expand[% - C2]
```

Out[125]=

$3 a^2 c + 6 a c t - 3 c (a + t)^2 + x$

Out[126]=

0

Out[127]=

$a c - c (a + t) + u$

Out[128]=

0

3 Скейлинг и P_2

КдФ допускает решения вида

$$u(x, t) = t^{-2/3} y(t^{-1/3} x), \quad y''' - 6 y y' + \frac{1}{3} (X y' + 2 y) = 0.$$

In[129]:=

```
sol2[y0_, y1_, y2_, Mx_] :=
 NDSolve[{y'''[X] == 6 y[X] × y'[X] - 1/3 (X y'[X] + 2 y[X]), y[0] == y0, y'[0] == y1,
    y''[0] == y2, WhenEvent[Abs[y[X]] > 100, "StopIntegration"]}, y, {X, -Mx, Mx}][[1]]

pl2[s_] := Plot[y[X] /. s, {X, -35, 35},
  PlotRange → {{-35, 35}, {-6.4, 2.4}},
  PlotStyle → Blue,
  AxesLabel →
   (Style[#, {FontSize → 20, FontFamily → "Times New Roman", FontSlant → Italic}] & /@
     {"X", "y"}),
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 400]
```

In[131]:=

```
Manipulate[s = sol2[y0, y1, y2, 100]; pl2[s],
 {{y0, -0.5}, -3, 3, Appearance → "Labeled"},
 {{y1, -0.6}, -3, 3, Appearance → "Labeled"},
 {{y2, -3}, -3, 3, Appearance → "Labeled"}
]
```

Out[131]=



In[132]:=

```
s = sol2[-0.5, -0.6, -3, 125];

g = Plot3D[t^(-2/3) y[t^(-1/3) x] /. s, {x, -15, 15}, {t, 0.1, 2},
  PlotRange → {{-15, 15}, {0.1, 2}, {-30, 5}},
  PlotPoints → 300,
  Mesh → None,
  AxesLabel → Evaluate[Style[#, Italic, 24] & /@ {"x", "t", "u"}],
  ImageSize → 600,
  BaseStyle → {FontFamily → "Times", FontSize → 16},
  ViewPoint → {0, -3, 1}
 ]
```

Out[133]=



Автомодельная подстановка приводит к P_{34} (с точностью до тривиальных замен)

In[134]:=

```
eqY = -y'''[X] + 6 y[X] × y'[X] - 1/3 X y'[X] - 2/3 y[X]
eqy = Expand[eqY /. {y[X] → y[X] + X/6, y'[X] → y'[X] + 1/6} /. y → v]
eqv = -v[X] × v''[X] + 1/2 v'[X]^2 + 2 v[X]^2 (v[X] + X/6) + c
Expand[eqy v[X] - D[eqv, X]]
```

Out[134]=

$$-\frac{2\,y[X]}{3}-\frac{1}{3}X\,y'[X]+6\,y[X]\,y'[X]-y^{(3)}[X]$$

Out[135]=

$$\frac{v[X]}{3}+\frac{2}{3}X\,v'[X]+6\,v[X]\,v'[X]-v^{(3)}[X]$$

Out[136]=

$$c+2\,v[X]^2\left(\frac{X}{6}+v[X]\right)+\frac{1}{2}v'[X]^2-v[X]\,v''[X]$$

Out[137]=

0

Преобразование Миуры к P_2

In[138]:=

```
eqy = Expand[eqY /. {y[X] → w'[X] + w[X]^2, y^(n_)[X] :→ D[w'[X] + w[X]^2, {X, n}]}]
eqw3 = -w'''[X] + 6 w[X]^2 w'[X] - 1/3 X w'[X] - 1/3 w[X]
Factor[eqy - D[eqw3, X] - 2 w[X] eqw3]

eqw = -w''[X] + 2 w[X]^3 - 1/3 X w[X] + a
eqw3 - D[eqw, X]
```

Out[138]=

$$-\frac{2}{3}w[X]^2-\frac{2\,w'[X]}{3}-\frac{2}{3}X\,w[X]\,w'[X]+12\,w[X]^3\,w'[X]+$$
$$12\,w[X]\,w'[X]^2-\frac{1}{3}X\,w''[X]+6\,w[X]^2\,w''[X]-2\,w[X]\,w^{(3)}[X]-w^{(4)}[X]$$

Out[139]=

$$-\frac{w[X]}{3}-\frac{1}{3}X\,w'[X]+6\,w[X]^2\,w'[X]-w^{(3)}[X]$$

Out[140]=

0

Out[141]=

$$a-\frac{1}{3}X\,w[X]+2\,w[X]^3-w''[X]$$

Out[142]=

0

Сравнение уравнений для *v* и *w*

In[143]:=

```
eqv
eqw

Expand[eqv /. {v[X] → w'[X] + w[X]^2 - X/6, v^(n_)[X] :→ D[w'[X] + w[X]^2 - X/6, {X, n}]}]
Expand[% /. Solve[{eqw, D[eqw, X]} == 0, {w''[X], w'''[X]}][[1]]]
```

Out[143]=

$$c + 2\,v[X]^2 \left(\frac{X}{6} + v[X]\right) + \frac{1}{2}\,v'[X]^2 - v[X]\,v''[X]$$

Out[144]=

$$a - \frac{1}{3}\,X\,w[X] + 2\,w[X]^3 - w''[X]$$

Out[145]=

$$\frac{1}{72} + c + \frac{1}{18}\,X^2\,w[X]^2 - \frac{2}{3}\,X\,w[X]^4 + 2\,w[X]^6 + \frac{1}{18}\,X^2\,w'[X] - \frac{1}{3}\,w[X]\,w'[X] - \frac{4}{3}\,X\,w[X]^2\,w'[X] + 6\,w[X]^4\,w'[X] - \frac{1}{3}\,X\,w'[X]^2 + 6\,w[X]^2\,w'[X]^2 - \frac{w''[X]}{6} + \frac{1}{3}\,X\,w[X]\,w''[X] - 2\,w[X]^3\,w''[X] + \frac{1}{2}\,w''[X]^2 + \frac{1}{6}\,X\,w^{(3)}[X] - w[X]^2\,w^{(3)}[X] - w'[X]\,w^{(3)}[X]$$

Out[146]=

$$\frac{1}{72} + c - \frac{a}{6} + \frac{a^2}{2}$$

4 sin-Gordon и P_3

$$u_{x\,t} = \sin 2\,u$$

In[147]:=

```
solu3[u0_, u1_, Mz_] := NDSolve[{u''[z] == -u'[z] / z + 1 / z Sin[2 u[z]], u[1] == u0, u'[1] == u1,
    WhenEvent[Abs[u[z]] > 1000, "StopIntegration"]}, u, {z, 0.0001, Mz}][[1]]

pl3[s_] := Plot[u[z] /. s, {z, 0, 100},
  PlotRange → {{0, 104}, {-2.75, 5.15}},
  PlotStyle → Blue,
  AxesLabel →
   (Style[#, {FontSize → 20, FontFamily → "Times New Roman", FontSlant → Italic}] & /@
     {"z", "y"}),
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 400]

Manipulate[s = solu3[u0, u1, 900];
 pl3[s],
 {{u0, 0.5}, -3, 3, Appearance → "Labeled"},
 {{u1, 1}, -3, 3, Appearance → "Labeled"}]
```

Out[149]=



In[150]:=

```
g = Plot3D[u[x t] /. s, {x, 0.01, 30}, {t, 0.01, 30},
  PlotRange → {{0, 30}, {0, 30}, {-5, 5.15}},
  PlotPoints → 300,
  Mesh → None,
  AxesLabel → Evaluate[Style[#, Italic, 24] & /@ {"x", "t", "u"}],
  ImageSize → 600,
  BaseStyle → {FontFamily → "Times", FontSize → 16}
 ]
```

Out[150]=



Регулярное решение. В окрестности 0 строим решение в виде ряда Тейлора, дальше продолжаем NDSolve

In[151]:=

```
Clear[z]

G[u_] = Sin[2 u];
Gu[u_] = D[G[u], u];
eps = D[G[u], u, u] / (4 G[u])

(*--- ряд Тейлора по z ---*)
seriesp[v0_, M_] := Module[{a, g, h, n, j},
  a[0] = v0;
  g[0] = G[v0];
  h[0] = Gu[v0];
  Do[a[n] = g[n - 1] / n^2;
   g[n] = Sum[k a[k] × h[n - k], {k, 1, n}] / n;
   h[n] = 4 eps Sum[k a[k] × g[n - k], {k, 1, n}] / n, {n, 1, M}];
  {Sum[a[n] z^n, {n, 0, M}], If[a[M] == 0, 100, Abs[a[M]]^(-1/M)]}
 ]

(*--- решение с начальным условием в точке z0≠0 ---*)
solvp[v0_, v1_, zmin_, z0_, zmax_] := Module[{s, err = If[z0 == zmin, zmax, zmin]},
  s = NDSolve[{z v''[z] + v'[z] == G[v[z]],
     v[z0] == v0, v'[z0] == v1,
     WhenEvent[Abs[v[z]] > 10, {err = z, "StopIntegration"}]},
    v, {z, zmin, zmax}];
  {err, s[[1]]}]

(*--- решение с начальным условием в точке z0=0 ---*)
sol0vz[v0_, M_, zmin_, zmax_] := Module[{V, Vz, r, nsl, nsr},
  V = seriesp[v0, M];
  r = 0.1 V[[2]];
  V = V[[1]];
  Vz = D[V, z];
  nsl = solvp[V /. z → -r, Vz /. z → -r, zmin, -r, -r];
  nsr = solvp[V /. z → r, Vz /. z → r, r, r, zmax];
  Piecewise[{
    {v[z] /. nsl[[2]], nsl[[1]] < z ≤ -r},
    {V, -r < z < r},
    {v[z] /. nsr[[2]], r ≤ z < nsr[[1]]}}]
 ]
```

Out[154]=

−1

In[158]:=

```
L = 20;
V = sol0vz[0.3, 100, -L^2, L^2];

gv = Plot[Evaluate[V], {z, -L^2, L^2}, PlotRange → {{-L^2, L^2}/5, {-0.5, 2.4}}, PlotStyle → Blue,
  AxesLabel →
   (Style[#, {FontSize → 20, FontFamily → "Times New Roman", FontSlant → Italic}] & /@
     {"z", "y"}),
  BaseStyle → {FontSize → 16, FontFamily → "Times New Roman"},
  ImageSize → 400]
```

Out[160]=



In[161]:=

```
g = Plot3D[V /. z → x t, {x, -L, L}, {t, -L, L},
  PlotRange → {{-L, L}, {-L, L}, {-0.5, 2.4}},
  PlotPoints → 400,
  Mesh → None,
  AxesLabel → Evaluate[Style[#, Italic, 24] & /@ {"x", "t", "u"}],
  ImageSize → 600,
  BaseStyle → {FontFamily → "Times", FontSize → 16}
 ]
```

Out[161]=



5 Тест Ковалевской-Пенлеве

$w'' = 2\,w^3 + z\,w + \alpha \; + \beta\,z$

Делаем замену $z - z_0 = Z$

In[162]:=

```
eq = -w''[z] + 2 w[z]^3 + (z + z0) (w[z] + β) + α

r = 7;
W = p / z + Sum[z^i c[i], {i, 0, r}]

zz = Collect[eq /. {w''[z] → D[W, z, z], w[z] → W}, z, Factor];
```

Out[162]=

$\alpha + 2\,w[z]^3 + (z + z0)\,(\beta + w[z]) - w''[z]$

Out[164]=

$\frac{p}{z} + c[0] + z\,c[1] + z^2\,c[2] + z^3\,c[3] + z^4\,c[4] + z^5\,c[5] + z^6\,c[6] + z^7\,c[7]$

Коэффициент p

In[166]:=

```
eqp = Expand[Coefficient[zz, z, -3]]
```

Out[166]=

$-2\,p + 2\,p^3$

Уравнение при z^{-2} определяет c_0

In[167]:=

```
eq0 = Factor[Coefficient[zz, z, -2]]
coefs0 = {c[0] → 0, p^2 → 1, p^3 → p};
Factor[eq0 /. coefs0]
```

Out[167]=

$6\,p^2\,c[0]$

Out[169]=

0

Уравнение при z^{-1} определяет c_1

In[211]:=

```
eq1 = Expand[Coefficient[zz, z, -1]] /. coefs0
coefs1 = Append[coefs0, c[1] → -p z0 / 6];
Factor[eq1 /. coefs1]
```

Out[211]=

$p\,z0 + 6\,c[1]$

Out[213]=

0

Уравнение при z^0 определяет c_2

In[173]:=

```
eq2 = Expand[Coefficient[zz, z, 0]] /. coefs1
coefs2 = Append[coefs1, c[2] → -(p + α + z0 β) / 4];
Factor[eq2 /. coefs2]
```

Out[173]=

$p + \alpha + z0\,\beta + 4\,c[2]$

Out[175]=

0

Уравнение при z^1: резонанс, c_3 не определяется, но появляется ограничение на β

In[176]:=

```
eq3 = Expand[Coefficient[zz, z, 1]] //. coefs2
coefs3 = Append[coefs2, β → 0];
Factor[eq3 /. coefs3]
```

Out[176]=

β

Out[178]=

0

Уравнение при z^2, z^3 определяет c_4, c_5 и т.д.

In[179]:=

```
eq4 = Expand[Coefficient[zz, z, 2]] //. coefs3
eq5 = Expand[Coefficient[zz, z, 3]] //. coefs3
```

Out[179]=

$-\frac{p\,z0}{6} - \frac{1}{4}\,z0\,(-p-\alpha) - 6\,c[4]$

Out[180]=

$-\frac{p\,z0^3}{108} + \frac{1}{4}\,(-p-\alpha) + \frac{3}{8}\,p\,(-p-\alpha)^2 - z0\,c[3] - 14\,c[5]$

Можно решать не пошагово, а сразу несколько уравнений

In[181]:=

```
eqp = Collect[Coefficient[zz, z, -3] / 2, c[_], Expand]
eq0 = Collect[Coefficient[zz, z, -2], c[_], Expand]
eq1 = Collect[Coefficient[zz, z, -1], c[_], Expand]
eq2 = Collect[Coefficient[zz, z, 0], c[_], Expand]

solc = Solve[({eq0, eq1, eq2} /. p^2 → 1) == 0, {c[0], c[1], c[2]}][[1]]
```

Out[181]=

$-p + p^3$

Out[182]=

$6\,p^2\,c[0]$

Out[183]=

$p\,z0 + 6\,p\,c[0]^2 + 6\,p^2\,c[1]$

Out[184]=

$p + \alpha + z0\,\beta + 2\,c[0]^3 + c[0]\,(z0 + 12\,p\,c[1]) + (-2 + 6\,p^2)\,c[2]$

Out[185]=

$\left\{c[0] \to 0,\ c[1] \to -\frac{p\,z0}{6},\ c[2] \to \frac{1}{4}\,(-p - \alpha - z0\,\beta)\right\}$

In[186]:=

```
eq3 = Collect[Coefficient[zz, z, 1], c[_], Expand]
eq4 = Collect[Coefficient[zz, z, 2], c[_], Expand]
eq5 = Collect[Coefficient[zz, z, 3], c[_], Expand]

Expand[eq3 /. solc /. {p^2 → 1, p^3 → p}]
```

Out[186]=

$$\beta + z0\, c[1] + 6\, c[0]^2\, c[1] + 6\, p\, c[1]^2 + c[0]\, (1 + 12\, p\, c[2]) + \left(-6 + 6\, p^2\right) c[3]$$

Out[187]=

$$6\, c[0]\, c[1]^2 + z0\, c[2] + 6\, c[0]^2\, c[2] + c[1]\, (1 + 12\, p\, c[2]) + 12\, p\, c[0] \times c[3] + \left(-12 + 6\, p^2\right) c[4]$$

Out[188]=

$$2\, c[1]^3 + c[2] + 6\, p\, c[2]^2 + z0\, c[3] + 6\, c[0]^2\, c[3] + c[1]\, (12\, c[0] \times c[2] + 12\, p\, c[3]) + 12\, p\, c[0] \times c[4] + \left(-20 + 6\, p^2\right) c[5]$$

Out[189]=

β

6 КдФ. WTC-тест

Ищем решение КдФ в виде ряда Лорана по $x - q(t)$. На первом шаге находим коэффициент при $(x - q)^{-2}$

In[190]:=

```
U = Sum[a_s[t] (x - q[t])^(s-2), {s, 0, 15}]
eq := Collect[-D[U, t] + D[U, x, x, x] - 6 U D[U, x] /. x → X + q[t], X, Factor]
cfeq[j_] := Coefficient[eq /. arule, X, j]

arule = {};
cfeq[-5]
```

Out[190]=

$$\frac{a_0[t]}{(x - q[t])^2} + \frac{a_1[t]}{x - q[t]} + a_2[t] + (x - q[t])\, a_3[t] + (x - q[t])^2\, a_4[t] + (x - q[t])^3\, a_5[t] + (x - q[t])^4\, a_6[t] + (x - q[t])^5\, a_7[t] + (x - q[t])^6\, a_8[t] + (x - q[t])^7\, a_9[t] + (x - q[t])^8\, a_{10}[t] + (x - q[t])^9\, a_{11}[t] + (x - q[t])^{10}\, a_{12}[t] + (x - q[t])^{11}\, a_{13}[t] + (x - q[t])^{12}\, a_{14}[t] + (x - q[t])^{13}\, a_{15}[t]$$

Out[193]=

$$12\, (-2 + a_0[t])\, a_0[t]$$

Следующие несколько уравнений позволяют найти a_1, a_2, a_3. Коэффициент a_4 остается свободным.

In[194]:=

```
arule = {a0[t] → 2};
cfeq[-5]
cfeq[-4]
cfeq[-3]
cfeq[-2]
cfeq[-1]
```

Out[195]=

0

Out[196]=

$30\, a_1[t]$

Out[197]=

$2\left(3\, a_1[t]^2 + 12\, a_2[t] - 2\, q'[t]\right)$

Out[198]=

$6\, a_1[t]\, a_2[t] + 12\, a_3[t] - a_1[t]\, q'[t] - a_0'[t]$

Out[199]=

$-a_1'[t]$

Далее находим a_5, а a_6 также свободен. Остальные находятся однозначно

In[200]:=

```
arule = {a0[t] → 2, a1[t] → 0, a2[t] → 1/6 q'[t], a3[t] → 0};
cfeq /@ {-5, -4, -3, -2, -1}
cfeq[0]
cfeq[1]
cfeq[2]
```

Out[201]=

$\{0, 0, 0, -a_0'[t], -a_1'[t]\}$

Out[202]=

$-6\, a_5[t] - a_2'[t]$

Out[203]=

$-a_3'[t]$

Out[204]=

$24\, a_7[t] - a_4'[t]$

In[205]:=

```
arule = {a_0[t] → 2, a_1[t] → 0, a_2[t] → 1/6 q'[t],
    a_3[t] → 0, a_5[t] → -1/36 q''[t], a_7[t] → 1/24 D[a_4[t], t]};

cfeq /@ {-5, -4, -3, -2, -1, 0, 1, 2}
cfeq[3]
cfeq[4]
cfeq[5]
Solve[{%%%, %%, %} == 0, {a_8[t], a_9[t], a_10[t]}]
```

Out[206]=

$$\left\{0,\ 0,\ 0,\ -a_0'[t],\ -a_1'[t],\ -a_2'[t]+\frac{q''[t]}{6},\ -a_3'[t],\ 0\right\}$$

Out[207]=

$$-12\, a_4[t]^2 + 72\, a_8[t] - a_5'[t]$$

Out[208]=

$$150\, a_9[t] - a_6'[t] + \frac{5}{6}\, a_4[t]\, q''[t]$$

Out[209]=

$$-36\, a_4[t]\, a_6[t] + 264\, a_{10}[t] - a_7'[t] - \frac{1}{72}\, q''[t]^2$$

Out[210]=

$$\left\{\left\{a_8[t] \to \frac{1}{72}\left(12\, a_4[t]^2 + a_5'[t]\right),\right.\right.$$
$$\left.\left. a_9[t] \to \frac{1}{900}\left(6\, a_6'[t] - 5\, a_4[t]\, q''[t]\right),\ a_{10}[t] \to \frac{2592\, a_4[t]\, a_6[t] + 72\, a_7'[t] + q''[t]^2}{19\,008}\right\}\right\}$$